Comsats Institute for Information Technology
Maths Department

April-May 2012
Stochastic Calculus and Finance

LASTNAME : Date : .
FIRSTNAME : Group : .

Answer-sheet n°1
Trajectories of an n time-steps CRR model

Please answer as clearly as possible in the provided space. The sheets will be collected at the end of the session.

Open Scilab and open an editor by clicking `Applications-Editor` in the top line of the `Console` window. Key-in the successive parts of the programme below (last page). Please realize that to have any part of the text to be executed by `Scilab`, you just need to highlight it and select `Execute-EvaluateSelection` in the menu of the Editor window.

Exercise 1. : Computation of the CRR random walk Recall that the CRR random walk S_t is defined from its initial value S_0 and the relation $S_{t+\delta t} = S_t U_t$, with $U_t \in \{\mathtt{up},\mathtt{down}\}$. Here we chose $up = e^{+\frac{\sigma}{\sqrt{n}}}$ and $down = e^{-\frac{\sigma}{\sqrt{n}}}$, where σ is a constant (the volatility of the asset) and n the number of time-steps, $T = n\delta t$. The various values of S_t are kept in a (triangular) matrix[1] $S(i,j) = SS(i+1, j+1)$, where $i = 0, \dots, n$ is the index of the time-step $t = i\delta t$, and $j = 0, \dots, i$ is the number of *up*s that took place up to time t.

Compute with `Scilab` the values of S_t in a model with $n = 3$ time-steps and draw on the provide grid the ten points with coordinates $(i, S(i,j))$ for $i = 0, \dots, 3$ and $j = 0, \dots, i$) and draw a line between those that can be reached from any time-step to the next one in order to draw the CRR tree of this elementary model. Remember that there is a unit-shift on i and j, as $S(i,j) = SS(1+i, 1+j)$.



What is the value of σ ?

[1] Numbering of lines and columns in Scilab begin at 1 and not 0 !

Exercise 2. : Plot of one trajectory Each trajectory is defined by a succession of `up`, `down` that we code by a sequence of 0 and 1. Let $i \mapsto J(i)$ be the function that gives the number of `up`s between $t = 0$ and $t = i\delta t$. So, for the trajectory associated with the sequence $1, 1, 0, 0, 0, 1, 0$, one has $J(2) = J(5) = 2$ and $J(6) = 3$.

1. Change the value of n (and recompute SS for the new value of n) and type the next part of the code. Please observe that the instruction `xset("window",1)` selects, and possibly creates window 1 for the forthcomming drawing graphic instructions (this window can be cleared with `clf(1)`). The code uses `cumsum`. Figure out what this produces, possibly using the text-editor help browser (highlight `cumsum` and hit `F1`-key or select " ?" in the menu).

2. Produce several trajectories by changing the sequence of zeros and ones and rerun the lines that generates the corresponding plot. What are the values of S for a sequence that alternatively has 0 and 1. Describe and explain the picture of this trajectory.

Exercise 3. : Simulated random sequence of 0 **and** 1

1. Function `rand()` of `Scilab` (just as the `random` key of a pocket-calculator) produces a random number between 0 and 1, uniformally distributed on $[0, 1]$. Using `rand(m,n)`, one gets an $m \times n$ matrix of numbers unifomally distributed on $[0, 1]$. Experiment with various (small) values of m and n. Function `2*rand()-1` also produces random numbers, but no longer in $[0, 1]$. Test it several times and explain what law is simulated in this way.

2. The two next instructions allow to check if `rand` simulates indeed a uniform law.
   `x=rand(1,1000);histplot(12,x)`
   Describe and explain what you observe ; what is the effect of the above `12`. What would you have expected ?

3. Possibly using the on-line help what is the purpose of function `int`. Check you answer by computing `int(-3.7)...`

4. Function `int(0.5+rand())` produces also a random number. What is the law of the random sequence simulated in this way ?

Exercise 4. : Simulation of M CRR trajectories

1. Now we wish to simulate M frajectories of the CRR random walk. For a given $p \in (0,1)$, what produces, `int(p+rand(n,M))`?

2. If we set `deltaJ=int(p+rand(n,M))`, what can you say of the result of `cumsum(deltaJ,"c")` ?

3. Let $n = 100$. Simulate $M = 40$ trajectories, choosing $p = 0.5$. Repeat this simulation using different values of the "volatility" σ, observe and describe the influence of this parameter on trajectories.

Exercise 5. : Probabilistic law of the last (T) value of a CRR random walk. Let $n = 200$. Draw the histogram of the final value of trajectories. To this purpose, simulate at least $M = 400$ trajectories and choose for example a number of classes equal to `int(sqrt(M))`. Describe the shape of the histogram you obtained.

```
  // Anything that comes after "//" in a line is just a comment
// that is not considered by Scilab.
// Please put as many comments as possible
// in your codes.
  //Exercise 1 : Computation of the CRR walk /////////////
  n=3 ;T=1 ;delta_t=T/n ;
  SS=zeros(n+1,n+1) ;
  S0=140 ;sigma=0.2 ;
  up=exp(sigma*sqrt(delta_t)) ;down=1/up ;
  SS(1,1)=S0 ;
  for i=1 :n
      SS(i+1,0+1)=SS(i,0+1)*down; // here j=0
      for j=1 :i
          SS(i+1,j+1)=SS(i,j)*up ;
      end ;
  end ;
  // Exercise 2 : Plot of a possible CRR trajectory : here n=10 ;
// Please rerun the definition of SS with n=10.

  deltaJ=[1 1 0 0 0 1 0 0 1 1] ;
  J=cumsum(deltaJ) ;
  ttraj=zeros(1,n+1) ;
  ttraj(1)=SS(1,1) ;
  for i=1 :n
      ttraj(i+1)=SS(i+1,J(i)+1) ;
  end ;
  xset("window",1) ;
  plot2d(0 :10,ttraj) ;
  // Exercise 3 : Simulation of a random sequence of zeros and ones ////////////////////////
  //rand() rand(3,2) rand(3,3) 2*rand()-1
  x=rand(1,1000)
  xset("window",2) ;
  histplot(12,x) ;
  int(0.5+rand(1,10))
  // Exercice 4 : Simulations of M CRR trajectories ////////////////
  M=100 ; n=100 ;
  MdeltaJ=int(0.5+rand(M,n)) ;
  MJ=cumsum(MdeltaJ,"c") ;
  Mttraj=zeros(M,n+1) ;
  for m=1 :M
      Mttraj(m,1)=S0 ;
      for i=1 :n
          Mttraj(m,i+1)=SS(i+1,MJ(m,i)+1) ;
      end ;
  end ;
  xset("window",3) ;
  for m=1 :M
      plot2d(0 :n,Mttraj(m,1 :n+1)) ;
  end ;
  // Exercice 5 : Distribution of the final value of a CRR walk ; //Chose M=400 ;
  xset("window",4) ;
  histplot(int(sqrt(M)),Mttraj(1 :M,n+1)) ;
```